
Appendix III

%MACRO LOGISTICFORWARD(DATA, RESPONSE, CLASSVARS, EFFECTS, SLENTRY=0.05);

%MACRO MATCHEFF(EFFECT1, EFFECT2);

%LOCAL II VAR;

%IF (%SYSFUNC(COUNTW(&EFFECT1)) NE %SYSFUNC(COUNTW(&EFFECT2))) %THEN 0;

%ELSE %DO;

%LET EFFECT1=%UPCASE(*&EFFECT1*);

%LET II=1; %LET VAR=%SCAN(&EFFECT2, &II);

%DO %WHILE (%INDEX(&EFFECT1, *&VAR*));

%LET II=%EVAL(&II+1);

%LET VAR=%SCAN(&EFFECT2, &II);

%END;

%IF ("&VAR"="") %THEN 1; %ELSE 0;

%END;

%MEND;

%MACRO REMOVEEFFECT(THISEFFECT, EFFECTS);

%LOCAL II VAR;

%LET THISEFFECT=%UPCASE(&THISEFFECT); %LET EFFECTS=%UPCASE(&EFFECTS);

%LET II=1; %LET VAR=%SCAN(&EFFECTS, &II, %STR( ));

%DO %WHILE ("&VAR" NE "");

%IF (NOT %MATCHEFF(&THISEFFECT, &VAR)) %THEN &VAR;

%LET II=%EVAL(&II+1); %LET VAR=%SCAN(&EFFECTS, &II, %STR( ));

%END;

%MEND;

%MACRO SELECTCLASSVARS(CLASSVARS, EFFECTS);

%LOCAL II VAR;

%LET II=1; %LET VAR=%SCAN(&EFFECTS, &II);



%DO %WHILE ("&VAR" NE "");

%IF %INDEX( %STR( %UPCASE(&CLASSVARS) ), %STR( %UPCASE(&VAR) )) %THEN &VAR;

%LET II=%EVAL(&II+1);

%LET VAR=%SCAN(&EFFECTS, &II);

%END;

%MEND;

%LOCAL II VAR THISCLASSVARS THISPVALUE THISEFFECT ENTERIT INCLUDEDEFFS;

%LET ENTERIT=1; %LET INCLUDEDEFFS=;

%DO %WHILE (&ENTERIT);

PROC SQL; CREATE TABLE THISSTEP (EFFECT CHAR(50), PROBCHISQ NUM);

%LET II=1; %LET VAR=%SCAN(&EFFECTS, &II, " ");

%DO %WHILE ("&VAR" NE "");

%LET THISCLASSVARS=%SELECTCLASSVARS(&CLASSVARS, &INCLUDEDEFFS &VAR);

PROC LOGISTIC DATA=&DATA DESCENDING;

%IF ("&THISCLASSVARS" NE "") %THEN CLASS &THISCLASSVARS/PARAM=GLM; ;

%IF ("&THISCLASSVARS" NE "") %THEN ODS OUTPUT MODELANOVA=_TYPE3;

%ELSE ODS OUTPUT PARAMETERESTIMATES=_TYPE3; ;

MODEL &RESPONSE=&VAR &INCLUDEDEFFS;

RUN;

%IF ("&THISCLASSVARS" EQ "") %THEN %DO; /*ALL CONTINOUS VARIABLES*/

DATA _TYPE3; SET _TYPE3 (RENAME=(VARIABLE=EFFECT)); IF(EFFECT NE "Intercept"); RUN;

%END;

DATA _TYPE3; SET _TYPE3; IF _N_=1;

PROC SQL; INSERT INTO THISSTEP SELECT "&VAR" AS EFFECT, PROBCHISQ FROM _TYPE3;

QUIT; RUN;

%LET II=%EVAL(&II+1);

%LET VAR=%SCAN(&EFFECTS, &II, " ");

%END;

PROC SQL; SELECT PROBCHISQ, (PROBCHISQ LE &SLENTRY) , EFFECT INTO

:THISPVALUE, :ENTERIT, :THISEFFECT FROM THISSTEP ORDER BY PROBCHISQ;



%PUT STEP - TRYING TO ADD %CMPRES(&THISEFFECT) WITH THE SMALLEST PVALUE OF

&THISPVALUE, ADDING FLAG: &ENTERIT;

%IF &ENTERIT %THEN %DO;

%LET EFFECTS = %REMOVEEFFECT(&THISEFFECT, &EFFECTS);

%LET INCLUDEDEFFS=&INCLUDEDEFFS &THISEFFECT;

%IF ("%TRIM(&EFFECTS)" EQ "") %THEN %LET ENTERIT=0;

%END;

%END;

%LET THISCLASSVARS=%SELECTCLASSVARS(&CLASSVARS, &INCLUDEDEFFS);

PROC LOGISTIC DATA=&DATA;

%IF ("&THISCLASSVARS" NE "") %THEN CLASS &THISCLASSVARS/PARAM=GLM; ;

MODEL &RESPONSE=&INCLUDEDEFFS;

RUN;

%MEND LOGISTICFORWARD;

/*TEST CODE******************************************************************

DATA SIMUDATA;

ARRAY X{*} X1-X20;

DO I=1 TO 500;

DO J=1 TO 20;

X{J}=NORMAL(-1)*2.5;

END;

IF UNIFORM(-1)<0.5 THEN C1=1; ELSE IF UNIFORM(-1)<0.5 THEN C1=2; ELSE C1=3;

IF UNIFORM(-1)<0.5 THEN C2=1; ELSE IF UNIFORM(-1)<0.5 THEN C2=2; ELSE C2=3;

XB=1+3*X1+2*X2-1*X3+5*(C1=1)+X1*X2+5*(C1=2)*(C2=2);

PROB=EXP(XB)/(1+EXP(XB));

Y=(UNIFORM(-1)>PROB);

OUTPUT;



END;

DROP I J XB PROB;

RUN;

%LOGISTICFORWARD(SIMUDATA, Y, C1 C2, X1 X2 X3 X4 X5 X6 X7 X8 X9 X10 X11 X12 X13 X14 X15

X16 X17 X18 X19 X20 C1 C2 X3*X1 X2*X1 C2*C1);

PROC LOGISTIC DATA=SIMUDATA;

CLASS C1 C2;

MODEL Y=X1 X2 X3 X4 X5 X6 X7 X8 X9 X10 X11 X12 X13 X14 X15 X16 X17 X18 X19 X20 C1 C2 X1*X2

X1*X3 C2*C1/SELECTION=FORWARD;

RUN;

******************************************************************************/


